

Graphic.wpm

Graphic.wpm is a WordPerfect macro that utilizes IrfanView (freeware Version 3.7 or later) and WordPerfect's command-line Graphic Conversion Utility GRAPHCNV.EXE (Version 1.1 or later) to provide WordPerfect 5.1 for DOS with compatibility for all modern image file types. In operation, graphic.wpm creates two temporary graphic files (c:\wp51\graphic.pcx and c:\wp51\graphic.wpg) and calls three batch programs (c:\wp51\graphic1.bat, c:\wp51\graphic2.bat, and c:\wp51\graphic3.bat).

Graphic.wpm

```
{;}GRAPHIC.WPM~

{;}delete temporary files~
{Shell}2del c:\wp51\graphic.pcx{Enter}{Enter}
{Shell}2del c:\wp51\graphic.wpg{Enter}{Enter}

{;}initialize variables~
{ASSIGN}conv~0~
{ASSIGN}rc~0~
{ASSIGN}refresh~0~

{;}reassign cancel to return and refresh screen~
{ON CANCEL}{GO}cancel~~

{;}check reveal codes~
{IF}{STATE}&512~{Reveal Codes}{ASSIGN}rc~1~{END IF}

{;}begin macro processing and request input path and file name~
{LABEL}begin~
{TEXT}filename~Enter graphic path and file name (? = help): ~

{;}check input length of path~
{IF}{LEN}filename~>34~{BELL}
{STATUS PROMPT}{Del to EOP}~{STATUS PROMPT}~{PROMPT}
This macro supports DOS paths/file names up to 34 characters only~
{WAIT}28~{ASSIGN}refresh~1~{GO}cancel~{END IF}

{;}cancel if no input~
{IF}"{VARIABLE}filename~"=""~{GO}cancel~~{END IF}

{;}check for requested help~
{IF}"{VARIABLE}filename~"="?"~

{;}help screen~
{LABEL}help~
{STATUS PROMPT}{Del to EOP}
{Enter}{Enter}{Enter}{Enter}{Enter}{Enter}
GRAPHIC.WPM HELP SCREEN{Enter}{Enter}
Enter the full DOS path and file name of the graphic file{Enter}
you want to insert into your document (e.g. a:\star06.gif,{Enter}
c:\clickart\sunrise.jpg, or c:\mydocu~1\signat~1.tif). Long{Enter}
file names can be quoted ("C:\My Documents\signature.tif").{Enter}
Only paths and file names up to 34 characters are supported.{Enter}
Next, choose your conversion option:{Enter}
  1.Grey scale - convert file to 16 shades of grey;{Enter}
  2.Black & white - convert file to black and white;{Enter}
  3.Color(256) - convert file to 256 colors; and{Enter}
  4.None(PCX) - used only with 8-bit PCX files.{Enter}
Two intermediate files named graphic.pcx and graphic.wpg will{Enter}
be stored in the WP51 directory during the conversion process.{Enter}
The intermediate files will remain available for renaming or{Enter}
copying until the next conversion process, at which time the{Enter}
stored intermediate files will be updated with the new graphic.{Enter}
{Enter}{Enter}
Press any key to continue. . . .~
{IF}{VARIABLE}conv~1~{GO}loop~{END IF}
{ASSIGN}refresh~1~
{CHAR}null~~
{ASSIGN}refresh~0~
{GO}begin~
```

```

{END IF}

{;}wait for conversion option, help request, or cancel key~
{LABEL}loop~
{ASSIGN}conv~1~
{CHAR}opt~Convert options:  1.Grey scale  2.Black & white  3.Color(256)  4.None(PCX) ~
{IF}"{VARIABLE}opt~"="1"~{ASSIGN}switch~/G=16~{GO}viewer~{END IF}
{IF}"{VARIABLE}opt~"="g"~{ASSIGN}switch~/G=16~{GO}viewer~{END IF}
{IF}"{VARIABLE}opt~"="G"~{ASSIGN}switch~/G=16~{GO}viewer~{END IF}
{IF}"{VARIABLE}opt~"="2"~{ASSIGN}switch~/C=2~{GO}viewer~{END IF}
{IF}"{VARIABLE}opt~"="b"~{ASSIGN}switch~/C=2~{GO}viewer~{END IF}
{IF}"{VARIABLE}opt~"="B"~{ASSIGN}switch~/C=2~{GO}viewer~{END IF}
{IF}"{VARIABLE}opt~"="3"~{ASSIGN}switch~/C=256~{GO}viewer~{END IF}
{IF}"{VARIABLE}opt~"="c"~{ASSIGN}switch~/C=256~{GO}viewer~{END IF}
{IF}"{VARIABLE}opt~"="C"~{ASSIGN}switch~/C=256~{GO}viewer~{END IF}
{IF}"{VARIABLE}opt~"="4"~{GO}none~{END IF}
{IF}"{VARIABLE}opt~"="n"~{GO}none~{END IF}
{IF}"{VARIABLE}opt~"="N"~{GO}none~{END IF}
{IF}"{VARIABLE}opt~"="?"~{GO}help~{END IF}
{GO}loop~

{;}shell to DOS calling graphic viewer (c:\wp51\graphic1.bat)~
{LABEL}viewer~
{IF}"{MID}filename~0~1~"="a"~{GO}floppy~{END IF}
{IF}"{MID}filename~0~1~"="A"~{GO}floppy~{END IF}
{Shell}2c:\wp51\graphic1 {VARIABLE}filename~ {VARIABLE}switch~{Enter}
{GO}post~

{;}additional pause for floppy (c:\wp51\graphic2.bat)~
{LABEL}floppy~
{Shell}2c:\wp51\graphic2 {VARIABLE}filename~ {VARIABLE}switch~{Enter}
{GO}post~

{;}shell to DOS without calling graphic viewer (c:\wp51\graphic3.bat)~
{LABEL}none~
{Shell}2c:\wp51\graphic3 {VARIABLE}filename~{Enter}

{;}post-conversion screen refresh processing~
{LABEL}post~
{Enter}
{STATUS PROMPT}{Del to EOP}~
{STATUS PROMPT}~
{Block}
{Home}{Up}
{Goto}{Goto}
{Home}{Down}
{Goto}{Goto}{Cancel}
{IF}{VARIABLE}rc~=1~{Reveal Codes}{END IF}
{Graphics}111c:\wp51\graphic.wpg{Enter}
{GO}end~

{;}cancel and screen refresh processing~
{LABEL}cancel~
{STATUS PROMPT}{Del to EOP}~
{STATUS PROMPT}~
{Block}
{Home}{Up}
{Goto}{Goto}
{Home}{Down}
{Goto}{Goto}{Cancel}
{IF}{VARIABLE}refresh~=1~{ASSIGN}refresh~0~{GO}begin~{END IF}
{IF}{VARIABLE}rc~=1~{Reveal Codes}{END IF}

{LABEL}end~

```

Graphic1.bat (store in C:\WP51; configure to "Close on Exit"):

```
@echo off
c:
cd\
cd progra~1
cd irfanv~1
i_view32.exe %1 /bpp=8 /convert=c:\wp51\graphic.pcx
cls
echo.
echo.
echo.
echo.
pause
cd\
cd wp51
dir *.pcx
graphcnv %2=%3 c:\wp51\graphic.pcx c:\wp51\graphic.wpg
```


Graphic2.bat (store in C:\WP51; configure to "Close on Exit"):

```
@echo off
c:
cd\
cd progra~1
cd irfanv~1
i_view32.exe %1 /bpp=8 /convert=c:\wp51\graphic.pcx
cls
echo.
echo.
echo.
echo.
echo Please wait until floppy drive activity has ceased and then
pause
cd\
cd wp51
dir *.pcx
graphcnv %2=%3 c:\wp51\graphic.pcx c:\wp51\graphic.wpg
```



Graphic3.bat (store in C:\WP51; configure to "Close on Exit"):

```
copy %1 c:\wp51\graphic.pcx
@echo off
c:
cd\
cd wp51
dir *.pcx
graphcnv c:\wp51\graphic.pcx c:\wp51\graphic.wpg
```



[NOTE: Graphic.wpm could be used with any graphic utility that can 1) run from a command line, 2) accept replaceable parameters (i.e. "%1") as a wildcard for the input file name, and 3) convert the input file to a PCX file with the name c:\wp51\graphic.pcx and a color depth limited to 256 or less. Only the three shaded lines in each of the first two batch files would need to be changed. It is possible that the batch programs do not need to be configured to "Close on Exit".]



>>>



>>>



>>>

